

MAB 471  
2011.2

# Análise Sintática - TINY

<http://www.dcc.ufrj.br/~fabiom/comp>



# EBNF

---

- Notação estendida para gramáticas
- Repetição
  - $\{...\} \rightarrow A ::= A \dots \mid \langle \text{vazio} \rangle$
- Opcional
  - $[...] \rightarrow A ::= \dots \mid \langle \text{vazio} \rangle$
- Repetição e opcional têm tradução direta em parsers recursivos
  - Loops e ifs



# Gramática EBNF para TINY

---

```
tiny          ::= cmd-seq
cmd-seq       ::= cmd {';' cmd}
cmd           ::= if-cmd | repeat-cmd | assign-cmd
               | read-cmd | write-cmd
if-cmd        ::= IF exp THEN cmd-seq
               [ELSE cmd-seq] END
repeat-cmd    ::= REPEAT cmd-seq UNTIL exp
assign-cmd    ::= ID ':=' exp
read-cmd      ::= READ ID
write-cmd     ::= WRITE exp
exp           ::= simple-exp [rel-op simple-exp]
rel-op        ::= '<' | '='
simple-exp     ::= term {add-op term}
add-op        ::= '+' | '-'
term          ::= factor {mul-op factor}
mul-op        ::= '*' | '/'
factor        ::= '(' exp ')' | NUMBER | ID
```



# Exemplo - Fatorial

---

```
{ programa de exemplo
  tiny - fatorial }

read x;
if 0 < x then {nao calcula de x menor que 0}
  fact := 1;
  repeat
    fact := fact * x;
    x := x - 1
  until x = 0;
  write fact { saida }
end
```



# Parser recursivo para TINY

---

Um tradutor de TINY para Lua ([www.lua.org](http://www.lua.org))

- Vamos usar o scanner gerado pelo JFlex
- Saída do parser será um programa Lua equivalente ao programa original TINY
- A semântica de TINY é simples o bastante que a tradução é bem direta
- Nosso parser vai abortar em casos de erros sintáticos